




# **EXTENSION RTC ALICE 4K - 32 - 90**




**FRED\_72**



## Extension RTC pour Alice

L'extension RTC (Real Time Clock) permet d'ajouter une horloge et un calendrier à votre ordinateur ALICE. L'extension est enfermée dans un boîtier qu'il suffit de brancher directement sur le bus d'extension situé à l'arrière de votre ordinateur.



L'extension n'occupe que 2 adresses 49000 et 49001 (\$BF68-BF69). L'horloge est alimentée par une pile longue durée qui permet de maintenir l'horloge à l'heure pendant plus de 10 ans. Le changement de date est automatique ainsi que le passage de l'heure d'été à l'heure d'hiver et inversement. Cette extension ajoute aussi 113 octets de mémoire RAM indirecte sauvegardée.

**Attention :** Pour éviter tout dommage à votre ordinateur ALICE, il faut impérativement débrancher l'alimentation avant de brancher l'extension.

## Structure mémoire de l'horloge

La mémoire de l'horloge (au total 128 octets) est accessible via 2 adresses. L'adresse 49000 (\$BF68) permet de sélectionner l'adresse mémoire de la RAM (0 à 127) de l'horloge tandis que l'adresse 49001 (\$BF69) permet de lire ou d'écrire dans cette mémoire. La structure mémoire est représentée sur la **Figure 1**.




Figure 1: Structure de la mémoire de l'horloge RTC

Les 14 premiers octets sont réservés à l'horloge ainsi que l'octet 50. Le reste de la mémoire peut être utilisé pour stocker des informations qui resteront mémorisées même si l'alimentation est coupée. La durée de vie de la batterie intégré au circuit horloge est d'environ 10 ans.

## Registres de l'horloge

L'horloge comporte 4 registres de configuration permettant de définir l'environnement matériel du circuit intégré. Une fois initialisés, il ne faut plus y toucher sous peine de voir l'horloge s'arrêter de fonctionner. Seuls les registres A et B peuvent être modifiés. Les valeurs par défaut sont :

Registre A = 43

Registre B = 15

**Remarque :** Les registres (Alarme) ne sont pas utilisés.

## Programmation

La programmation se fait à l'aide des commandes POKE et PEEK du BASIC ou bien en langage machine.

L'utilisation est simple, Il suffit de lire les registres de l'horloge conformément à la **Figure 1** pour obtenir les informations d'heure, de minutes, et de secondes.

Pour cela, il faut d'abord écrire le N° du registre à l'adresse 49000 puis ensuite lire l'adresse 49001 pour obtenir la valeur du registre.

Les heures, minutes, secondes, la date et le mois sont stockés au format binaire. Les valeurs ne nécessitent aucun post-traitement. Les années sont séparées en 2, le siècle stocké dans l'octet 50 (19 ou 20) et les dizaines et unités stockées dans l'octet 9.

Le jour de la semaine est numérique : 1 correspond à dimanche tandis que 7 correspond à samedi.

## Mise à l'heure de l'horloge

Ce programme BASIC permet de mettre l'horloge à l'heure.

```
10 PRINT "MISE A L'HEURE RTC"
20 INPUT "JOUR [1-7] :";J
30 INPUT "DATE [1-31] :";D
40 INPUT "MOIS [1-12] :";MO
50 INPUT "ANNEE [0-99] :";A
60 INPUT "HEURE [0-23] :";H
70 INPUT "MINUTE [0-59] :";M
80 INPUT "SECONDE [0-59] :";S
100 REM TRANSFERT VERS RTC
110 POKE 49000,11:POKE 49001,143
120 POKE 49000,50:POKE 49001,20
130 POKE 49000,9:POKE 49001,A
140 POKE 49000,8:POKE 49001,MO
150 POKE 49000,7:POKE 49001,D
160 POKE 49000,6:POKE 49001,J
170 POKE 49000,4:POKE 49001,H
180 POKE 49000,2:POKE 49001,M
190 POKE 49000,0:POKE 49001,S
200 POKE 49000,11 : POKE 49001,15
210 PRINT "MISE A JOUR TERMINEE"
220 PRINT
230 END
```

Une fois l'horloge réglée, l'ordinateur peut être débranché tout comme l'interface et lors de la prochaine utilisation, il suffira de lire l'heure. Il est inutile de la régler à nouveau.

## Lecture de l'horloge

Ce programme BASIC permet de lire l'heure et la date à partir de l'horloge RTC.

```
10 PRINT "LECTURE DE L'HEURE"  
20 POKE 49000,9: A=PEEK(49001)  
30 POKE 49000,8: MO=PEEK(49001)  
40 POKE 49000,7: D=PEEK(49001)  
60 POKE 49000,4: H=PEEK(49001)  
70 POKE 49000,2: M=PEEK(49001)  
80 POKE 49000,0: S=PEEK(49001)  
100 PRINT "NOUS SOMMES LE ";D;"/";MO;"/20";A  
110 PRINT "IL EST ";H;":";M;":";S
```

**Remarque :** Les composants DS12C887+ disponibles sur Aliexpress ou ebay sont souvent de mauvaise qualité avec notamment un quartz peu précis. Il est alors normal que l'horloge dérive de manière parfois importante. Il est possible de remplacer ce quartz mais c'est une manœuvre très délicate. Il est donc plus sage de s'approvisionner auprès d'un vendeur connu proche de chez nous.

## Schéma du matériel

